
Tipi di Variabili
Tipi semplici:
	  
	Valore di Default

	integer
	(numero intero Es: 0,1,2, ........)
	0

	real
	(numero reale Es: 0.1,0.2,.........)
	0.0

	boolean
	(valore booleano Es: true, false)
	false

	character
	(variabile simbolica Es: a,A,b,B,......)
	iso null (i.e. invisibile)

	text
	(array di simboli Es: ciccio, pippo,...)
	no text (i.e. "")


Tipi complessi
Si ottengono per composizione dei tipi semplici utilizzando il costrutto Class che ha una funzione molto simile a quella del costrutto Struct / Class del C / C++.
Dichiarazione delle variabili
Il modello di dichiarazione di una variabile è il seguente:
TipoVariabile Identificatore1, Identificatore2, .......IdentificatoreN, .......;
Es:
integer NumeroStudenti; !dichiariamo la variabile NumeroStudenti di tipo intero; 
text NomeStudente; !dichiariamo la variabile NomeStudente di tipo array di simboli;
Oss:
Si noti che ogni istruzione deve terminare con il simbolo ; e che eventuali commenti possono essere inseriti dopo il simbolo ! terminandoli come al solito con ;. Occorre infine ricordare di moderare la lunghezza del nome di un identificatore e che questa non deve mai iniziare con un carattere numerico.
Istruzione di assegnazione
Il modello più semplice di istruzione di assegnazione è il seguente:
Identificatore1 := Identificatore2;
Es:
real a; a := 1/2; !dichiariamo la variabile a di tipo reale e le assegnamo il valore 1/2;
Oss: Un discorso a parte deve essere invece fatto per le variabili di tipo puntatore
Es:
ref(TipoVariabile) P,Q; !P e Q sono variabili puntatore al tipo complesso TipoVariabile;
P :- Q; !P e Q condividono la stessa locazione di memoria;
Input/Output
La tabella successiva mostra le principali funzioni per l'I/O di differenti tipi di variabili.
	
	Input
	Output

	variabile intera
	InInt
	OutInt(NomeVariabile,m)

	variabile reale
	InReal
	OutReal(NomeVariabile,n,m)

	variabile simbolica
	InChar
	OutChar(NomeVariabile)

	variabile array di simboli
	InText(s)
	OutText(puntatore)


con m che è la dimensione del campo da usare quando il valore della variabile verrà visualizzato, n è il numero di cifre dopo la virgola che devono essere vusializzate, ed s è la lunghezza massima di caratteri che può contenere l'array di simboli. 
!-----------------------------------------------------------------------------------
 Semplice programma che mostra l'I/O di un numero intero.
 -----------------------------------------------------------------------------------;
begin   !si noti che ogni programma Simula inizia sempre con begin e termina con end;
   integer a;

   OutImage; !riempie la riga corrente di spazi e va a capo, intuitivamente è come il ritorno carrello;
   OutText("Inserisci un numero intero: ");

   BreakOutImage; 

   a := InInt; OutImage; 

   OutText("Il numero intero inserito e': ");

   OutInt(a,6) 

end
!--------------------------------
 Prova di esecuzione.
 --------------------------------;
Inserisci un numero intero:     12
Il numero intero inserito e':     12
!----------------------------------------------------------------------------------
 Semplice programma che mostra l'I/O di un numero reale.
 ----------------------------------------------------------------------------------;
begin
   real a;

   OutImage; !riempie la riga corrente di spazi e va a capo, intuitivamente è come il ritorno carrello;
   OutText("Inserisci un numero reale: ");

   BreakOutImage;

   a := InReal; OutImage;

   OutText("Il numero reale inserito e': ");

   OutFix(a,2,6) 

end


!--------------------------------
 Prova di esecuzione.
 --------------------------------;
Inserisci un numero reale:   3.2
Il numero reale inserito e':   3.20
!----------------------------------------------------------------------------
 Semplice programma che mostra l'I/O di un simbolo.
 ----------------------------------------------------------------------------;
begin
   character a;

   OutImage; !riempie la riga corrente di spazi e va a capo, intuitivamente è come il ritorno carrello;
   OutText("Inserisci un simbolo: ");

   BreakOutImage; a := InChar; OutImage;

   OutText("Il simbolo inserito e': ");

   OutChar(a) 

end
!---------------------------------
 Prova di esecuzione.
 ---------------------------------;
Inserisci un simbolo: F
Il simbolo inserito e': F
!-------------------------------------------------------------------------------------------
 Semplice programma che mostra l'I/O di una stringa di simboli.
 -------------------------------------------------------------------------------------------;
begin
   text a;

   OutImage; !riempie la riga corrente di spazi e va a capo, intuitivamente è come il ritorno carrello;
   OutText("Inserisci una parola: "); BreakOutImage;

   a :- InText(20); !è un'assegnazione a un riferimento, infatti a è un puntatore vero e proprio;
   OutImage; OutText("La parola inserita e': ");

   OutText(a) 

end
!---------------------------------
 Prova di esecuzione.
 ---------------------------------;
Inserisci una parola: gatto
La parola inserita e': gatto
Operatori aritmetici
La tabella successiva mostra i principali operatori aritmetici:
	+
	addizione

	-
	sottrazione

	*
	moltiplicazione

	**
	elevamento a potenza

	/
	divisione tra reali

	//
	divisione tra interi


Oss:
Nel caso della divisione tra interi, gli operandi devono necessariamente essere interi; se il risultato dell'operazione comporta dei decimali questi verranno ignorati.
Connettivi
 

La tabella successiva mostra i principali connettivi:
	 per tipi semplici
	=
	>
	<
	>=
	<=

	per puntatori
	= =
	=/=

	per tipi booleani
	eqv
	not
	and
	or
	  

	concatenazione di aray di simboli
	&


 Procedure & Funzioni
Il modello più semplice di procedura è il seguente: 

(TipoVariabile) procedure NomeProcedura (Parametro1,..., ParametroN,...);

TipoVariabile Parametro1,..., ParametroN,...;

begin
   ...;

end

	Tipo
	Valore
	Referimento
	Nome

	tipi semplci
	default
	non lecito
	lecito

	text
	lecito
	default
	lecito


!----------------------------------------------------------------------------------------------------------

 Procedura che somma due numeri interi A e B e restituisce il risultato in C.

 ----------------------------------------------------------------------------------------------------------;
begin
   procedure Add(A,B,C); name C; integer A, B, C;
   begin
      C := A + B

   end of Add;

   integer ValA, ValB, ValC;   

   ValA := 1; ValB := 2; ValC := 0;

   Add(ValA,ValB,ValC); OutImage;OutInt(ValA,1); 

   OutText(" + "); OutInt(ValB,1); OutText(" = "); OutInt(ValC,1)

end
!---------------------------------
 Prova di esecuzione.

----------------------------------;
1 + 2 = 3
!-------------------------------------------------------------------
 Funzione che somma due numeri interi A e B.
 -------------------------------------------------------------------;
begin
   integer procedure Sum(A,B); integer A, B; !i parametri sono passati per valore;
   begin
      Sum := A + B
   end of Sum;

   integer ValA, ValB, ValC;

   ValA := 1; ValB := 2; ValC := 0; ValC := Sum(ValA,ValB); OutImage;

   OutInt(ValA,1); OutText(" + "); OutInt(ValB,1);

   OutText(" = "); OutInt(ValC,1)

end
!---------------------------------

 Prova di esecuzione.

 ---------------------------------;
1 + 2 = 3
Selezione & Iterazione
Le principali istruzioni di selezione e iterazione sono riassunte nella seguente tabella:
	 If
	While
	For


Istruzione IF
 

Il comando if del Simula si serve della parola chiave if più o meno come si usa in italiano la parola "se". La frase "se il bar è aperto mi porti una birra, altrimenti un caffè" ha la stessa stesura in Simula:
if BAR = APERTO then BEV := BIRRA else BEV := CAFFE';
Quando si usa il commando if la sintassi prevede che il then sia seguito da una istruzione. Se il programma prevede un altro if questo deve essere racchiuso da un blocco begin-end come segue:
if Condizione1 then

begin
     if Condizione2 then Statement

end
Istruzione While
 

In un programma Simula possiamo specificare un ciclo mediante un istruzione while.
La forma del comando while è la seguente:
while Condizione do Statement;
 

Statement è il corpo del ciclo while; viene valutata prima la condizione, se questa è falsa, l'esecuzione del comando while è terminata mentre se è vera Stantement viene eseguito un'altra volta; la condizione controllata nuovamente, e cosi via.
Istruzione For 

Quando vogliamo eseguire più volte un'istruzione e il numero di ripetizioni non dipende dal risultato di istruzioni interne al ciclo, la costruzione più adatta è il ciclo For. La sintassi del comando For è la seguente: 
for I := 1 step +1 until 100 do Statement;
La variabile I assume inizialmente valore1, viene eseguito stantement, incrementata di 1 fino a che avrà valore 100, quindi il ciclo si fermerà solo in quel caso, abbiamo altre possibilità di for per esempio: 
for I := 100 step -1 until 1 do Statement;

for I := 1, I + 3 step -1 until 1 do Statement;

for I := 2, 3, 5, 7, 11 step  4 until 51 do Statement;

for C := 'A', 'E', 'I', 'O', 'U', 'Y' do Statement;

for P :- P1, P2, P3 do Statement;
Un nuovo concetto di CLASSE
Il Simula 67 fu definito inizialmente per far fronte a quello che era il limite dell'Algol 60, cioè la descrizione e la simulazione di sistemi discreti. Un sistema discreto è caratterizzato dalla particolarità che i suoi cambiamenti di stato hanno luogo a distanza di tempo comunque spaziati; nel simulare un sistema del genere, più che il contenuto delle attività reali, hanno rilevanza i loro effetti e gli istanti di tempo in cui occorrono gli eventi di inzio e di fine delle attività. 
Per gestire questo tipo di problemi, il Simula 67 introdusse il concetto di CLASSE, che diede origine all'idea di trattare dei "tipi didati astratti", cioè la possibilità di "parametrizzare" le astrazioni definite dall'utente, fornendo sostanzialmente, solamente l'effetto che esse generano e gli istanti di tempo in cui avvengono, non interessandosi dello specifico tipo di dato trattato, che potrebbe essere indifferentemente, per esempio, un insieme di numeri interi oppure una coda di clienti. 
Sebbene la creazione del Simula 67 era legata al superamento di questi problemi, proprio l'universalità del concetto di classe lo fece imporre come strumento per progettare programmi organizzati su diversi livelli di astrazione.
Dichiarazione di una Classe
Una classe è intesa come un normale blocco di programma, che comprende dichiarazioni, operazioni, chiamate a procedura o ad altre classi. Essa va dichiarata nella lista di dichiarazioni che appare nell'intestazione del programma, insieme alle variabili e alle procedure.Una dichiazione di classe ha la seguente forma :
< intestazione > ;
< corpo > ;
l'intestazione contiene il nome della classe e i parametri formali,
il corpo contiene dichiarazioni di variabili locali, istruzioni eseguibili, procedure.
Utilizzo di una Classe
Un esemplare di classe deve essere generato da un'istruzione NEW, che crea un oggetto già inizializzato, poichè il corpo di una classe è già eseguito automaticamente dall'istruzione New. Un esempio può chiarire meglio queste nozioni.Una volta costruita una classe, per utilizzarla si crea una variabile di tipo classe e ci si riferisce ad i suoi oggetti solo tramite un puntatore. Per esempio nelle istruzioni :
Ref (complex) c ; c :- New complex (1.0,1.0) ;
c è un puntatore a complex che è un oggetto creato dinamicamente nella seconda istruzione, in maniera tale di sortire l'effetto :
    raggio [ 0.78 ] <------------- c         angolo [ 1.42 ]           x [ 1.0 ]           y [ 1.0 ]
così alla fine le istruzioni c.raggio e c.angolo; forniranno rispettivamente i valori di raggio e angolo (cioè le componenti in forma polare del numero complesso) calcolati dalla classe complex.
Esempio di classe 
Un numero complesso in forma polare può essere descritto dalla seguente dichiarazione di classe, x e y denotano le sue componenti in forma cartesiana e angolo e raggio denotano le sue componenti in forma polare.
begin

      class complex(x,y); real x,y;

      begin

           real epsilon, angolo, raggio, pi;

           epsilon := 1/(10**16); !epsilon indica un valore reale usato come approssimazione dello zero;
           pi := 3.14;  !pi contiene il valore di pi-greco;
           raggio := sqrt(x**2+y**2); !sqrt è la funzione radice quadrata;
           if abs(x) < epsilon then 

           begin 

               if abs(y) < epsilon then OutText("Errore") 

               else            

               begin 

                   if y > epsilon then angolo := pi/2 else angolo := 3*pi/2

               end

           end

           else angolo := arctan(y/x)  !arctan è la funzione arcotangente;
      end complex;
      ref(complex) c;

      real v1, v2;
      OutImage; OutText("Inserire le componenti in forma cartesiana del numero"); OutImage;

      OutText("Inserisci il valore dell'ascissa: "); BreakOutImage; v1 := InReal; OutImage;

      OutText("Inserisci il valore dell'ordinata: "); BreakOutImage; v2 := InReal; OutImage;

      c :- new complex(v1,v2); OutImage;

      OutText("Le coordinate in forma polare sono: ");

      OutImage;

      OutText("Raggio: ");

      OutFix(c.raggio,2,6);

      OutImage;

      OutText("Angolo: ");

      OutFix(c.angolo,2,6)

end

!-------------------------------
 Prova di esecuzione
 -------------------------------;
Inserire le componenti in forma cartesiana del numero
Inserisci il valore dell'ascissa:2

Inserisci il valore dell'ordinata: 2

Le coordinate in forma polare sono:

Raggio: 2.83

Angolo: 0.79
Simulazione in Simula
La caratteristica principale del Simula che lo rende maggiormente adatto alla simulazione di processi discreti è la possibilità di definire un insieme di processi che operano concettualmente in parallelo e che mediante una routine di temporizzazione detta sequencing set (calendario degli eventi) si sincronizzano secondo lo schema delle attività da essi rappresentate (avanzamento del tempo per eventi). 
Una attività in simula è definita come un insieme di dichiarazioni di dati ed una sequenza di statement che descrivono il comportamento di una componente del sistema.Ogni attività può produrre più processi, uno per ogni assegnazione possibile di valori agli elementi della sua struttura dati. E' quindi possibile definire dinamicamente ad ogni istante nuovi processi, uno per ogni assegnazione possibile di valori agli elementi della struttura dati. 
L'insieme di tutti i processi di un dato tipo costituiscono una particoleare classe. Gli oggetti di una classe rappresentano i processi relativi ad una data simulazione.
Meccanismi di avanzamento del tempo
II comportamento di un sistema dinamico (discreto o continuo) è generalmente studiato rappresentando gli stati del sistema come funzioni del tempo, e raccogliendo ed analizzando poi le statistiche ottenute. Nella realtà le attività delle varie componenti del sistema si svolgono con simultaneità e questa condizione non è generalmente riproducibile nella simulazione perché il simulatore (ed in particolare il programma di simulazione) può prestare attenzione solo ad una componente del sistema alla volta.
Quindi quelle attività che nella realtà sono parallele nel programma di simulazione sono sequenziali ed il simulatore si interessa solo degli effetti di queste attività e degli istanti
di tempo in cui tali attività hanno inizio e fine. 
Gli eventi che cambiano lo stato di un sistema sono generati in diversi istanti di tempo ed il passaggio da un tempo ad un altro è rappresentato da un orologio interno al sistema (internal clock) che è incrementato dal simulatore. 
Il tempo simulato può essere incrementato in due modi:
· avanzamento per intervalli: l'internal clock è incrementato da un tempo t ad un tempo t+dt, con dt incremento fisso di tempo;
· avanzamento per eventi: l'internal clock è incrementato da un tempo t ad un prossimo tempo-evento t' (evento imminente rispetto a quello in t). 

Il metodo di avanzamento per intervalli prefissati di tempo è utilizzato nella simulazione di sistemi continui o nel caso di sistemi con un grande numero di variabili di stato. Questo metodo ha lo svantaggio che se dt è grande il simulatore attraversa diversi periodi di inattività in cui non avviene alcuna variazione negli stati del sistema.
Nel caso dell'avanzamento per eventi solo gli eventi sono esplicitamente rappresentati nel modello di simulazione ed i periodi tra due eventi sono considerati inattivi o insignificanti e quindi non consumano tempo (sebbene questo non sia vero nel sistema reale). Il tempo è quindi avanzato solo quando è previsto un evento nel sequencing set scavalcando tutti i periodi durante i quali non avvengono cambiamenti nel sistema.
Oltre al tempo simulato si possono distinguere altri due concetti di tempo
· il tempo reale;
· il tempo di esecuzione (computer time) 

Il tempo reale è quello proprio del sistema reale da simulare ed è una variabile continua.
Il tempo di esecuzione è il tempo consumato dall'elaboratore per l'esperimento di simulazione. In genere dipende dal numero e dalla complessità dei cambiamenti di stato che avvengono nel sistema, e non dalla scala degli intervalli di tempo simulati.
Meccanismi di avanzamento del tempo
II comportamento di un sistema dinamico (discreto o continuo) è generalmente studiato rappresentando gli stati del sistema come funzioni del tempo, e raccogliendo ed analizzando poi le statistiche ottenute. Nella realtà le attività delle varie componenti del sistema si svolgono con simultaneità e questa condizione non è generalmente riproducibile nella simulazione perché il simulatore (ed in particolare il programma di simulazione) può prestare attenzione solo ad una componente del sistema alla volta.
Quindi quelle attività che nella realtà sono parallele nel programma di simulazione sono sequenziali ed il simulatore si interessa solo degli effetti di queste attività e degli istanti
di tempo in cui tali attività hanno inizio e fine. 
Gli eventi che cambiano lo stato di un sistema sono generati in diversi istanti di tempo ed il passaggio da un tempo ad un altro è rappresentato da un orologio interno al sistema (internal clock) che è incrementato dal simulatore. 
Il tempo simulato può essere incrementato in due modi:
· avanzamento per intervalli: l'internal clock è incrementato da un tempo t ad un tempo t+dt, con dt incremento fisso di tempo;
· avanzamento per eventi: l'internal clock è incrementato da un tempo t ad un prossimo tempo-evento t' (evento imminente rispetto a quello in t). 

Il metodo di avanzamento per intervalli prefissati di tempo è utilizzato nella simulazione di sistemi continui o nel caso di sistemi con un grande numero di variabili di stato. Questo metodo ha lo svantaggio che se dt è grande il simulatore attraversa diversi periodi di inattività in cui non avviene alcuna variazione negli stati del sistema.
Nel caso dell'avanzamento per eventi solo gli eventi sono esplicitamente rappresentati nel modello di simulazione ed i periodi tra due eventi sono considerati inattivi o insignificanti e quindi non consumano tempo (sebbene questo non sia vero nel sistema reale). Il tempo è quindi avanzato solo quando è previsto un evento nel sequencing set scavalcando tutti i periodi durante i quali non avvengono cambiamenti nel sistema.
Oltre al tempo simulato si possono distinguere altri due concetti di tempo
· il tempo reale;
· il tempo di esecuzione (computer time) 

Il tempo reale è quello proprio del sistema reale da simulare ed è una variabile continua.
Il tempo di esecuzione è il tempo consumato dall'elaboratore per l'esperimento di simulazione. In genere dipende dal numero e dalla complessità dei cambiamenti di stato che avvengono nel sistema, e non dalla scala degli intervalli di tempo simulati.
Classe Simulation
Le procedure per costruire un modello di simulazione sono contenute nella classe di libreria detta class SIMULATION.
L’avanzamento del tempo attuato dalla class SIMULATION è quello per eventi e il tempo simulato è contenuto nella variabile TIME. 
Quando l’occorrenza di un evento associato ad un oggetto è stata prevista, si dice che l’oggetto è stato schedulato e gli si associa un tempo che è appunto quello di occorrenza dell’evento, che rappresenta l’istante di tempo in cui è previsto l’inizio della prossima fase attiva dell’oggetto. Quando la fase attiva termina l’oggetto può essere rischedulato o deschedulato.
La deschedulazione dell’oggetto avviene quando:
· tutte le sue azioni sono terminate;
· non si conosce l’istante iniziale della sua prossima fase attiva. 

Il primo oggetto eseguito è quello cui è associato il tempo di occorrenza più basso e
l’oggetto attualmente eseguito è detto oggetto corrente.
Per schedulare e deschedulare gli oggetti si utilizzano le procedure
· Activate 

· Passivate 

· Hold 

I soli oggetti considerati presenti nel sequencing set sono quello attivo e quelli sospesi. Tutti gli altri oggetti sono o passivi o terminati.
Per eliminare un oggetto x della lista degli eventi si usa il comando 
Cancel (x) ;
questa istruzione rende l’oggetto x passivo.
Ogni componente del sistema da simulare deve essere costruito sotto forma di classe ed incorporato nella class SIMULATION mediante il prefisso PROCESS . Ogni volta che si crea un nuovo oggetto questo entra a far parte dell’insieme degli oggetti passivi e qui aspetta il comando di attivazione da parte di un altro oggetto.
Se la creazione e l’attivazione sono eventi successivi si usa il comando 
Activate new nomeoggetto (link ad activate ed a new).
Un oggetto il cui corpo è stato completamente eseguito fino alla parola END del suo blocco più esterno è normalmente considerato terminato a meno che:
· l’ultimo statement prima dell’end sia un passivate e quindi l’oggetto resta in un stato passivo; 

· il corpo dell’oggetto è incluso in un ciclo while nel qual caso l’oggetto resta eseguibile finche la condizione è vera, (se simultime è il tempo di simulazione un esempio di condizione è Time<simultime che mantiene il processo eseguibile per tutta la durata della simulazione). 

Classe Process 
Usando il prefisso PROCESS è possibile inserire una classe nella più ampia classe SIMULATION. 
Alcuni attributi della classe PROCESS sono inaccessibili ai programmi Simula, in particolare non è possibile vedere l’event notice, cioè le procedure ti temporizzazione degli eventi che stilano il sequencing set.
Gli attributi visibili dell’oggetto PROCESS danno informazioni sullo stato corrente dell’oggetto e sono:
· Activate (rende un processo primo nel sequencing set); 

· Reactivate (rischedula un processo) 

· Suspended (lascia un processo nel sequencing set, ma non come primo della lista eventi); 

· Passive (elemina un processo che non ha ancora raggiunto l’END finale dal sequencing set. Il processo però può essere rischedulato). 

· Terminated (elimina definitivamente un proceso dal sequencing set perchè questo ha raggiunto l’END finale e non è possibile rischedularlo. 

· Idle (procedura booleana) 

Generazione di numeri casuali
Il Simula fornisce numerose procedure per la generazione di numeri pseudo-casuali aventi determinate distribuzioni:
· variabili aleatorie con distribuzione uniforme; 

· variabili aleatorie con distribuzione esponenziale; 

· variabili aleatorie con distribuzione di Poisson; 

· variabili aleatorie con distribuzione k-erlangiana: 

· variabili aleatorie con distribuzione normale; 

· generazione discreta di numeri interi. 

